
Introduction 
 
Depuis quelques années, l’utilisation du DSP (Digital Signal Processor) dans les 

systèmes électroniques devient indispensable, par exemple dans les chaînes hi-fi, les 
téléphones portables etc… . L’implantation  d’un DSP dans un produit de bien de 
consommation devient même un argument de vente par les constructeurs. Alors DSP 
phénomène de société ou véritable évolution dans le domaine des microprocesseurs ? Ainsi 
nous verrons exactement ce qu’est un DSP. En effet à l’heure du tout numérique nous verrons 
le rôle du DSP au sein des nouvelles technologies notamment celles des télécommunications. 
Nous nous intéresserons plus particulièrement au traitement d’une modulation AM via le DSP 
de chez Texas Instrument : le TMS 320C50. Après avoir justifier le choix de cette 
modulation, nous étudierons les méthodes employées pour sa réalisation , sa mise en œuvre 
ainsi que les différents essais. 

 
 

I- Présentation générale 
 
 Dans cette partie, nous allons étudier brièvement le DSP. Nous expliquerons la 
modulation d’amplitude, nous justifierons le choix de notre sujet par rapport au DSP et nous 
en exposerons les objectifs.  
 

1) Qu ‘est ce que le DSP 
 

DSP signifie Digital Signal Processor. Ce processeur est tout particulièrement adapté 
au traitement numérique du signal, domaine dans lequel un microprocesseur classique ou un 
microcontrôleur n’excelle pas. En effet en général les microprocesseurs décomposent les 
opérations de multiplications ou de divisions en une série de décalages, d’additions ou de 
soustractions. Le DSP pour sa part possède des fonctions spécifiques au traitement du signal   
(par exemples fonctions de multiplications ) qui le rende très efficace dans la rapidité de 
calcul. C’est en outre pour cela qu’il est utilisé dans les domaines aussi diverses que la 
télécommunication mobile, les systèmes de guidage, les radars anticollisions, mais aussi 
l’électroménager.  

En plus de sa rapidité, le DSP présente plusieurs avantages parmi lesquels on peut 
citer le fait que toutes les fonctions qui sont réalisées ne nécessitent aucun câblage et par 
conséquent aucun phénomène de vieillissement ou de dégradation n’est observé. Un autre 
avantage est l’obtention à chaque itération du processus, d’ un résultat tout à fait identique. 

Pour développer notre projet nous avons utilisé le Starter Kit de chez Texas Instrument 
au sein duquel est implanté le DSP TMS 320C50. Ce DSP est fabriqué en une technologie 
CMOS intégrée. La rapidité et la flexibilité, ses principales caractéristiques, sont le résultat 
d’une combinaison d’une architecture Harvard avancée ( séparation du Bus programme et Bus 
de données) et d’une CPU avec une logique matérielle pour des applications spécifiques. Le 
TMS 320C50 est désigné pour exécuter 50 millions d’instructions par seconde (MIPS). 

Sur le Kit on trouve également un circuit Convertisseur Analogique/Numérique, 
Numérique/analogique (TLC32040), un port série RS232 qui sera relié à l’ordinateur, et un 
convertisseur AC/DC 220-9V pour son alimentation. ( voir Annexe I, photo du kit) 

 Le convertisseur AN/NA est un convertisseur 14 bits de résolution. Il est constitué 
d’un filtre passe bande, et d’un filtre anti-repliement, qui limitent la fréquence de conversion à 
un maximum de 70Khz et restreignent la bande passante . 

Du point de vue logiciel, le kit est fourni avec un compilateur linker (DSK5A) et un 
debugger (DSK5D). 

 

2) Le sujet 
 

Nous avons donc opté pour la réalisation d’une modulation AM analogique via le 
DSP. Nous allons ainsi dans un premier temps faire quelques rappels en ce qui concerne la 
modulation d’amplitude et ensuite justifierons le choix de ce type de processeur. 

 
a. La modulation d’amplitude 

 
Les applications de la modulation d’amplitude se situent essentiellement dans le 

domaine des transmissions d’informations par voie hertzienne. Il existe cependant d’autres 
secteurs d’utilisation comme celui de la mesure où pour s’affranchir de problèmes de bruit en 
basse fréquence, on fait appel à la détection synchrone qui découle de la modulation AM. 

 La modulation du point de vue du traitement du signal se traduit par 
l’expression suivante :  
 

                                                              
)cos())(1( ttfmA ×Ω××+×

 

avec :  A : amplitude du signal porteur 
             m : taux de modulation  
            f(t) : signal informatif que l’on prendra sinusoï dal pour des commodités  

de résultat 
           )cos( tΩ  : signal porteur 
 
Concrètement la modulation applique un décalage au signal informatif autour d’une fréquence 
porteuse qui donne en pratique les oscillogrammes temporel et fréquentiel suivants :  

 
 
 

 
   
 
 

Porteuse 

info info 

b. Pourquoi le DSP pour la modulation AM ? 
 

Comme toute modulation, celle d’amplitude d’un signal analogique nécessite un 
traitement mathématique. Or nous savons que de part ses caractéristiques, vues 
précédemment, le DSP est adapté à ce type de traitement. Qui plus est, la modulation AM 
nécessite 2 sources : une source informative et une source sinusoï dale à la fréquence porteuse. 
Or le DSP nous permet de faire abstraction de l’oscillateur, en effet il est aisé par 
programmation de réaliser un oscillateur sinusoï dal quasi parfait isolé de tout bruit 
électronique extérieur. Il sera ensuite facile de modifier la fréquence de la porteuse en jouant 
sur les registres de configuration du DSP, opération beaucoup plus délicate sur un oscillateur 
de type analogique. 

Enfin dernier avantage, avec le DSP nous sommes certains d’obtenir un signal de 
sortie invariant au cours du temps, nous sommes à l’abris de toutes dérives de composants 
constituant un modulateur analogique. 
  
 

II- Réalisation. 

1) Principe. 
 

Le principe consiste à entrer sur Analog In du kit un signal analogique informatif dans 
le DSP, de générer simultanément un signal porteur et enfin d’appliquer le traitement adéquat 
dans le but d’obtenir un signal modulé en amplitude sur la sortie Analog out du kit. 

Le traitement se fait par 2 types d’opérations : multiplications et additions. Le signal 
informatif est dans un premier temps multiplié par le taux de modulation, le signal résultant 
est ensuite multiplié par un échantillon de porteuse et enfin à ce résultat est ajouté ce même 
échantillon de porteuse.  

On peut ainsi décomposer le traitement en différentes étapes sous la forme d’un arbre 
programmatique. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
     

  

Acquisition du signal informatif   
    f(t)   

Multiplication par le taux de modulation m   

Multiplication par le signal porteur  ) cos( t Ω     

Addition de ce signal porteur    

m*f(t)   

m*f(t)* ) cos( t Ω 
    

) cos( )) ( 1 ( t t f m × Ω × × + 
  

Génération de la porteuse ) cos( t Ω     

2) Génération de la porteuse. 
 
Deux méthodes s’offrent à nous pour la réalisation de la sinusoï de de la porteuse: 

- la table d’échantillon 
- la relation de récurrence . 

 
 

a. Génération de la porteuse à l’aide d’une table 
d’échantillons. 

  
On décompose la fonction porteuse en une série d’échantillon par période. On range 

ces échantillons dans un tableau de valeur. Ensuite, à l’aide du buffer circulaire spécifique au 
DSP 320c50, on sélectionne un échantillon différent à chaque interruption ( à la fréquence de 
conversion préalablement définie). Après le dernier échantillon, le buffer circulaire se 
réinitialise au début du tableau .     
 
 

b. Génération de la porteuse par récurrence. 
 

On utilise la méthode de la transformée en z. On part de la fonction mathématique de 
la porteuse  

 
 

)***2cos()( tfty π=
 

Et par la transformée, on obtient une équation récursive de type  
 

)2()1()1()()(
2110

−+−+−+= nynynxnxny bbaa  

 
(démonstration dans l’annexe II) 
avec : 

1

)2cos(2

)2cos(

1

2

1

1

0

−=

=

−=

=

b
b
a
a

ft

ft

π

π
 

 
Pour programmer cette équation , nous devons utiliser 3 tables et 3 buffers circulaires pour 
x(n-1) , y(n-1) et y(n-2). 

 
 

c. Notre choix. 
 

     Nous avons décidé de prendre la première solution. En effet, la génération par récurrence 
donne un signal porteur de meilleur qualité mais nécessite un temps d’exécution beaucoup 
plus important. De plus, cette récurrence nécessite 3 buffers circulaires or nous n’avons que 2 
buffers à notre disposition. Enfin la programmation est fastidieuse et n’est pas le thème 
principal de notre sujet. 

 Afin d’exploiter les capacités maximum du DSP nous avons fixé une fréquence 
d’échantillonnage de 69,120 kHz en initialisant TA à 5 et TB à 15. 
 

KHz
MHz

TBTA

f
f master

conv 12,69
1552

368.10

2
=

××
=

××
=  

  
Ces valeurs de TA et TB fixe les deux fréquences de coupure du filtre : la haute est 

fixée à 12,96 kHz et la basse à 1,08kHz. 
 

KHz
TAKHz

f
f master

cb 08,1300
2288

=×
××

=  KHz
TAKHz

f
f master

cb 96,123600
2288

=×
××

=   

 
 

Afin de respecter le théorème de Shannon ( fe>2*fp), la fréquence porteuse doit être 
égale au maximum à 34,56kHz. Nous fixerons cette fréquence porteuse à 5kHz . Cela signifie 
qu’une période du signal porteur sera composée de 14 échantillons ( 70 Khz / 5 Khz) , 
relativement suffisant pour reproduire une sinusoï de d’assez bonne qualité. Les échantillons 
seront pris d’amplitude bien inférieure à l’unité pour éviter certains problèmes liés au format 
Q15  
   

Pour la détermination de ces échantillons, nous avons utilisé le logiciel Matlab . 
Voici le table d’échantillons     
 
   1 2 3 4 5 6 7 
0.125 0.112625 0.077937 0.027815 -0.027812 -0.077937 -0.11262 
 
8 9 10 11 12 13 14 
-0.125 -0.112625 -0.077937 -0.027812 0.027812 0.077937 0.112625 
 
  Pour la courbe et le programme Matlab (voir en Annexe III). 
 
La génération de la porteuse se fait en plusieurs étapes : 
 
*insertion du tableau de valeurs, nommé « valeur.txt » dans la zone de données à l’aide de la 
commande « include ». 
 
.ds     0f30h               
.include "valeur.txt" 
 
  *initialisation du buffer circulaire : 
 

On charge l’adresse de début du tableau dans le registre  CBSR1 et l’adresse de fin 
dans CBER1. On initialise le buffer circulaire en mettant dans CBCR la valeur 0fh. Les trois 
bits de poids faibles sont égaux à 1 et correspondent à l’activation du registre 7 comme 
registre courant . Le quatrième 1 active le buffer numéro 1, et enfin les quatre bits de poids 
fort égaux à zéro impliquent que le buffer 2 est désactivé. On charge dans AR7 la valeur qui 
correspond au début du tableau. 

 SPLK    #valeur,CBSR1      
 SPLK    #valeurfin,CBER1                          
 SPLK    #0fh, CBCR      
 LAR     AR7,#valeur       
 

3) Multiplication des  signaux. 
 

  Afin de reconstituer le signal modulé, nous allons procéder en trois étapes :on 
multiplie le signal informatif par le taux de modulation, puis le résultat est multiplié par le 
signal porteur, et enfin on ajoute la porteuse. 
 
 

a. le taux de modulation. 
 

Il se trouve en zone de données au format Q15 une variable nommée « m ».Sa valeur 
varie de 0 à 0.99. 
En réalité, « m » peut être supérieur à 1 et dans ce cas on obtient surmodulation, mais le 
format choisi ne nous permet pas de dépasser l’unité. 

On récupère la donnée du signal informatif stocké dans le registre « DRR :Data 
Register Receive » à l’aide de la commande « LT ». Elle est sauvegardée dans le registre 
« TREG0 ». 
On charge ensuite « m » et grâce à la commande « MPY », on multiplie « m » par TREG0. 
 
LT DRR 
LDP #m 
MPY m 
 

On sauvegarde le résultat de la multiplication dans l’accumulateur par l’instruction 
« APAC », en ayant préalablement initialisé l’accumulateur à 0. Afin de convertir le résultat 
de la multiplication 32 bits en 16 bits, on réalise un décalage vers la droite de 15 bits . 
 
APAC 
RPT #14 
SFR 
    

Finalement, on mémorise la valeur résultante du décalage grâce à une variable 
temporaire nommée  « TEMP », placée en zone de données. 
 
LDP #TEMP 
SACL TEMP 
 

b. Multiplication de ce signal par la porteuse. 
 

On multiplie le résultat précédent par le contenu du registre AR7  qui contient l’un des 
14 échantillons du tableau de valeurs.  

On active le registre  AR7 par la ligne de commande « MAR *,AR7 » .On charge 
ensuite « TREG0 » par « TEMP » grâce à la commande « LT ». On effectue ensuite  la 
multiplication de AR7 et TREG0 . Comme pour la multiplication précédente, on stocke cette 

valeur  dans l’accumulateur  par l’ « APAC » en ayant eu soin de mettre ce dernier à zéro 
juste avant cette opération. On décale également de 15 bits ce résultat. 
 
MAR *,AR7 
LT TEMP 
MPY *                 
APAC 
RPT #14              
SFR 
 

c. Ajout du signal porteur. 
   
On peut réaliser cette opération par deux méthodes. 
 
 

i. Utilisation de la propriété du mnémonique « APAC ». 
 

L’  « APAC » réalise en principe le chargement dans l’accumulateur par sa dernière 
valeur incrémentée de PREG0. Si on charge l’accumulateur préalablement par un échantillon 
de la porteuse contenu dans AR7, on additionne le résultat de la multiplication avec la 
porteuse. 
On obtient donc le signal modulé. 

Pour cela , on charge dans l’accumulateur le contenu du registre actif (AR7), décalé de 
15bits ( Conversion 32à16). On met dans TREG0 la donnée TEMP. On multiplie enfin 
TREG0 et TEMP et on incrémente le registre AR7. On termine par l’addition  de 
l’accumulateur  et de PREG0 mémorisé dans l’accumulateur. 
 
LACC *,15 
LT TEMP 
MPY *+   
APAC      

 
ii. Ajout par l’instruction « and ». 

 
Suite à l’APAC, l’accumulateur contient le résultat  de m*f(t)* )cos( tΩ . Ensuite, on 

utilise la commande « ADD » pour réaliser l’addition entre l’accumulateur et le registre 
courant. On incrémente AR7 en préparation de l’interruption suivante. 
 
 ADD *+ 

4) La sortie du signal modulé. 
 

Le TLC32040 est un convertisseur numérique analogique de 14 bits de résolution. Il 
est donc nécessaire de masquer notre donnée de sortie et donc la convertir sur 14 bits au lieu 
de 16. Physiquement, les deux premiers bits de poids faible ne sont pas reliés au CAN. Pour le 
masquage, on utilise la ligne de  commande « and  0fffch » qui réalise un  « et » logique.  
AND #0fffch 
On charge le signal résultant vers « DXR » (registre de transmission de données). 
SACL DXR 
Programme disponible en Annexe IV. 

III- Résultats 
 
  
 L'objectif de cette partie est d'analyser le fonctionnement de notre programme  en 
effectuant diverses expériences telles que le changement du taux de modulation, de la 
fréquence porteuse et de la fréquence modulante. 
 

1) Essai. 
 
 Dans un premier temps ,nous avons vérifié la cohérence de notre signal de sortie après 
traitement par le DSP. Nous obtenons sans se préoccuper des paramètres d'entrée le résultat 
temporel suivant. 
 

 
 

On constate que l'on obtient bien une réponse fidèle à nos attentes, un signal sinusoï dal 
à la fréquence porteuse enveloppé par la sinusoï de du signal informatif. Par contre avec cette 
courbe nous ne pouvons pas voir si la modulation se fait avec ou sans porteuse. Pour cela 
nous avons effectuer l'analyse spectrale de ce même signal et on obtient le spectre suivant. 

La présence de la raie centrale nous prouve que le signal observé possède les propriétés d'une 
modulation AM avec porteuse. 

 
 
 

a. Evolution du taux de modulation 
 
 Nous savons que la variation du taux de modulation entraîne une évolution  
d'amplitude du signal modulé plus ou moins importante. Nous avons effectué 5 mesures avec 
les taux de modulation  suivants : m=0.01 ;  0.25 ; 0.5 ; 0.8 ; 0.99 
(Voir en Annexe V). 
Nous observons expérimentalement l'évolution prévue. Notamment sur les points extrêmes. 
Pour m =0.01 on obtient un signal très faiblement modulé, proche de la sinusoï de porteuse. Et 
pour m=0.99 on obtient une modulation maximum à la limite de la surmodulation. 
 

b. Variation du signal informatif. 
 

 
  Nous avons vérifié également que la variation de fréquence du signal informatif ne 
modifie pas les propriétés de la modulation. Pour cela nous nous sommes fixés 2 fréquences 
de signal modulant : 500 Hz et 1KHz contenues dans la zone de fonctionnement que nous 
définirons dans la partie suivante. 
Voir les réponses temporelles à chaque fréquence informative en Annexe VI. 
 Nous constatons que la fréquence de l'enveloppe évolue bien en fonction du signal 
appliqué par le générateur (porteuse). 
 

c. Variation du signal porteur. 
 

 
Nous pouvons également faire varier la fréquence du signal porteur. Il existe deux 

possibilités pour réaliser ce changement:  
 

- En modifiant les registres TA et TB. 
- En augmentant le nombre d’échantillons. 

Il faut savoir qu’en jouant sur les registres TA et TB, on modifie la fréquence de 

conversion ainsi que celle de coupure du filtre de CAN. 
En utilisant cette méthode , nous avons réalisé 4 mesures avec les fréquences porteuses 
suivantes: 3.34 KHz , 2.5 KHz, 1.8KHz et 1.2KHz 
(Voir courbe en Annexe VII). 
  

2) Les limites de fonctionnement. 
 
 Le Kit DSP n'est pas idéal, en effet plusieurs facteurs l'empèchent de fonctionner dans 
n'importe quelles conditions. Tout d'abord le filtre passe bande placé sur le TLC32040 
(convertisseur) possède une fréquence de coupure basse de 1.08 KHz et de coupure haute de 
12,96 Khz pour une fréquence de conversion de 70 KHz. Par conséquent on ne pourra  en 
aucun cas dépasser une fréquence maximale de spectre de 12.96 KHz ( 12.96 KHz > 
Fporteurse +Fmodulant ). Idem en ce qui concerne la fréquence minimum du spectre fixée à 
1.08KHz ( 1.08KHz< Fporteuse - Fmodulant). 
En dehors de cette bande spectrale le signal de sortie sera atténué. 
 
 L’intervalle de conversion du DSP est limitée : on ne pourra donc pas dépasser la 
fréquence de conversion maximale de 70 kHz. Ce qui limitera, suite au théorème de Shannon 
évoqué précédemment, la fréquence porteuse à un maximum de Fconv/2. 

 
 

Conclusion : 
 

L’étude de cette modulation nous a permis de découvrir le DSP à travers quelques 
fonctions de bases comme les tableaux, les multiplications, les buffers circulaires…. Il est 
clair que technologiquement, le DSP n’est jamais utilisé pour la réalisation de modulation 
d’amplitude, le rapport efficacité/coût comparé à un modulateur de type analogique justifie 
cela. Néanmoins ce sujet nous ouvre les portes des nouvelles modulations de type numérique 
utilisées en télécommunication (QPSK,QAM…). Ces modulations nécessitent un traitement 
mathématique plus élaboré par conséquent plus en adéquation avec l’utilisation d’un DSP. 

 ANNEXE II : Calcul de la porteuse par récurrence 

 
Transformée en z de la fonction tty ωcos)( =   
A tty ωcos)( = on associe Tnny ωcos)( =  où ω représente la pulsation du signal sinusoidal 
souhaité et T la période du signal d’échantillonnage. 

La transmittance en z à obtenir s ‘écrit : )(

1

)(

)(

)(
)( zY

zY

zX

zY
zH ===  

Pour un signal sinusoidal sougaité et une période d’échantillonnage définie les termes 
Tωsin et Tωcos sont des constantes. 

On pose : C= Tωsin , A=2 Tωcos , D= Tωcos , B=-1 
 

)1)(()1)((
1

1

1²

²

1cos2²

cos²

)(

)( 121
21

1
−−−

−−

−

−=−−→
−−

−
=

+−
−

=
+−

−
= DzzXBzAzzY

BzAz

Dz

Azz

Dzz

Tzz

Tzz

zX

zY

ω
ω

 
)()()()()( 121 zXDzzXzYBzzYAzzY −−− −=−−  
)1()()2()1()( −−=−−−− nDxnxnBynAyny  
)2()1()1()()( −+−+−−= nBynAynDxnxny  

 
)2()1()1()()( 2110 −+−+−+= nybnybnxanxany   

 
avec BbAbDaa ==−== 2110 ,,,1  

ANNEXE III : Calcul sous Matlab 
 
 
 
Signal Porteur F=5Khz et Fe = 70KHz (69.120 KHz) 
 

Programme Matlab 
 
» F=5000; 
» Fe=70000; 
» t=(0:13)/Fe; 
» x=(1/8)*cos(2*pi*t*F); 
» plot(x) 
 

Echantillons 
 
Columns 1 through 7  
 
    0.1250    0.1126    0.0779    0.0278   -0.0278   -0.0779   -0.1126 
 
Columns 8 through 14  
 
   -0.1250   -0.1126   -0.0779   -0.0278    0.0278    0.0779    0.1126 
 

Graphique du signal porteur échantillonné 

ANNEXE IV : Programme 
 

1) Modulation Amplitude par addition de porteuse à l’aide de 
la commande ADD. 

**************************************************************** 
*   (C) COPYRIGHT TEXAS INSTRUMENTS, INC. 1992             * 
**************************************************************** 
*  MODULE NAME: Modulation d’amplitude par APAC                * 
*                             * 
*  DESCRIPTION: This routine initializes the TMS320C50 and     * 
*               TLC32040 on the C5x DSK. Modulation d’amplitude* 
*               avec porteuse ( FP = 5 Khz )                   * 
*                                                              * 
*                             * 
**************************************************************** 
         .mmregs 
* 
;---------------------------------------------------------------; 
; Bit definition of the control register in the AIC             ; 
;---------------------------------------------------------------; 
 
      ;+-------------------------+ 
      ;|LP xx G1 G0 | SY AX LB BP|              G1  G0  gain 
      ;+------------+------------+             --------------- 
      ; |     GAIN    |  |  |  +-- BP Filter    0   0    4 
      ; |     Synch --+  |  +----- Loopback     1   1    4 
      ; |     Auxin -----+                      0   1    2 
      ; + (sinx)/x filter                       1   0    1 
 
 
         .ds     0f00h            ; 
TA       .word    5               ; Fconv = 69.12 KHz 
RA       .word    18              ; Fcut = 8 KHz 
TAp      .word    31              ; 
RAp      .word    31              ; 
TB       .word    15              ; Fs = 2*Fcut 
RB       .word    18              ; Fs = 2*Fcut 
AIC_CTR  .word    8h 
;valeur du taux de modulation 
m        .q15     0.25 
;variable temporaire 
TEMP     .word 0 
;tableau de valeur des échantillons 
         .ds     0f30h               

    .include "valeur.txt"      ;introduction des valeurs des  
 ;échantillons 

        
 
 
**************************************************************** 
*   Set up the ISR vector                                      * 
**************************************************************** 
         .ps     080ah 
rint:    B       RECEIVE       ;0A; Serial prot receive interrupt RINT. 
xint:    B       TRANSMIT      ;0C; Serial port transmit interrupt XINT. 
 
* 

****************************************************************** 
* TMS32C05X INITIALIZATION                                       * 
*         * 
****************************************************************** 
         .ps     0a00h 
         .entry 
START:   SETC    INTM             ; Disable interrupts 
         LDP     #0               ; Set data page pointer 
         OPL     #0834h,PMST 
         LACC    #0 
         SAMM    CWSR             ; Set software wait state to 0 
         SAMM    PDWSR            ; 
 
* Reset AIC by writing to PA2 (address >52) to DSK 
 
         SPLK    #022h,IMR        ; Using XINT syn TX & RX 
         CALL    AICINIT          ; initialize AIC and enable interrupts 
* 
******************************************************************* 
* This routine enables serial port rx interrupts & configures   * 
* TLC32040 for the frame sync. When RINT is triggered, read a     * 
* dummy data word from the AIC then generate a sine wave to       * 
* send out.                                                       * 
******************************************************************* 
; 
         CLRC    OVM              ; OVM = 0 
         SPM     0                ; PM = 0 
         SPLK    #012h,IMR 
         CLRC    INTM             ; enable 
 
   
;initialisation du buffer circulaire 
;on range dans CBSR l'adresse du début de tableau(0000 1111) 
;les 3 derniers chiffres servent à sélectionner le registre 7 
;le quatrième chiffre sélectionne l activation du buffer 1 
 
         LDP     #0 
         SPLK    #valeur,CBSR1    ;stock l’adresse de départ dans  
         ;l’accumulateur 
         SPLK    #valeurfin,CBER1 ;stock la valeur de l accumulateur: 
                                  ;adresse de fin de buffer 
         SPLK    #0fh, CBCR       ;configuration du buffer circulaire: 
                                  ;buffer 1 actif et on travaille avec AR7 
         
         LAR     AR7,#valeur      ;Chargement de AR7 par le début de valeur 
         MAR     *,AR7 
 
 
WAIT     NOP                      ; wait for a transmit or receive 
         NOP 
         NOP 
         B       WAIT 
 
;------- end of main program ----------; 
 
; 
; RECIEVER INTERRUPT SERVICE ROUTINE 
; 
RECEIVE: 
;activation du registre AR7 
         ZAP                     ;remise à zéro de l'accumulateur 

         LDP #0                  ;on se place en page zéro  
         LT DRR                  ;on met dans le registre TREG0 la valeur 
          ;de l'entrée du DSP 
         LDP #m      ;on charge le taux de modulation 
   
         MPY m                   ;on multiplie le registre TREG0 par m 
         APAC      ;on ajoute la valeur de l'accumulateur avec  
        ;le résultat de la multiplication 
         RPT #14                ;on décale à droite de 15 bits 
         SFR 
         LDP #TEMP               ;on charge TEMP 
         SACL TEMP               ;on met la valeur de l'accumulateur dans   
                                 ;TEMP 
      
;multiplication des signaux (porteuse et signal informatif)  
         ZAP                     ;mise a zéro de l accumulateur 
         LT TEMP                 ;on met TEMP dans TREG0 
         MPY *                   ;multiplication de DRR par AR7  
         APAC                    ;on additionne la valeur de l'accumulateur 
                                 ;et le résultat de la multiplication 
         RPT #14                 ;décalage de 16 bits (adaptation 32 16)  
         SFR       
         ADD *+                  ;on incrémente le registre AR7 après  
                                 ;l'avoir ajouté à l'accumulateur 
         AND #0fffch             ;Masquage de l’ACC pour conversion à 14  
                                 ;Bits pour le CNA 
         LDP #0                  ;on se place en page zéro 
         SACL DXR                ;sortie vers le registre de transmission 
         
   RETE                    ; end of program 
 
 
* 
; 
; TRANSMIT INTERRUPT SERVICE ROUTINE 
; 
TRANSMIT: 
         RETE 
 
* 
****************************************************************** 
*  DESCRIPTION: This routine initializes the TLC320C40 for       * 
*  a 8Khz sample rate with a gain setting of 1  * 
****************************************************************** 
* aic initialization data 
* 
AICINIT: SPLK    #20h,TCR           ; To generate 10 MHz from Tout 
         SPLK    #01h,PRD           ; for AIC master clock 
         MAR     *,AR0 
         LACC    #0008h             ; Non continuous mode 
         SACL    SPC                ; FSX as input 
         LACC    #00c8h             ; 16 bit words 
         SACL    SPC 
         LACC    #080h            ; Pulse AIC reset by setting it low 
         SACH    DXR 
         SACL    GREG 
         LAR     AR0,#0FFFFh 
         RPT     #10000           ; and taking it high after 10000 cycles 
         LACC    *,0,AR0          ; (.5ms at 50ns) 
         SACH    GREG 
         ;------------------------ 

         LDP     #TA              ; 
         SETC    SXM              ; 
         LACC    TA,9             ; Initialized TA and RA register 
         ADD     RA,2             ; 
         CALL    AIC_2ND          ; 
         ;------------------------ 
         LDP     #TB 
         LACC    TB,9             ; Initialized TB and RB register 
         ADD     RB,2             ; 
         ADD     #02h             ; 
         CALL    AIC_2ND          ; 
         ;------------------------ 
         LDP     #AIC_CTR 
         LACC    AIC_CTR,2        ; Initialized control register 
         ADD     #03h             ; 
         CALL    AIC_2ND          ; 
         RET                      ; 
 
AIC_2ND: 
         LDP     #0 
         SACH    DXR              ; 
         CLRC    INTM 
         IDLE 
         ADD     #6h,15           ; 0000 0000 0000 0011 XXXX XXXX XXXX XXXX 
b 
         SACH    DXR              ; 
         IDLE 
         SACL    DXR              ; 
         IDLE 
         LACL    #0               ; 
         SACL    DXR              ; make sure the word got sent 
         IDLE 
         SETC    INTM 
         RET                      ; 
 .end 
 

 
 

2) Modulation AM avec addition de porteuse à l’aide de la 
commande APAC . 

 
Partie utile du Programme . 
 
;------- end of main program ----------; 
 
; 
; RECIEVER INTERRUPT SERVICE ROUTINE 
; 
RECEIVE: 
;activation du registre AR7 
         ZAP 
         LDP #0                    
         LT DRR 
         LDP #m 
         MPY m 
         APAC 
         RPT #14 
         SFR 

         LDP #TEMP 
         SACL TEMP 
      
;multiplication des signaux (porteuse et signal informatif)  
         ZAP                  ;mise a zéro de l accumulateur 
         LACC *,15  ;Chargement de l’échantillon porteur dans 

;l’ACCU 
         LT TEMP 
         MPY *+               ;multiplication de DRR par AR7 et incrémente 
                                  ;AR7 
         APAC   ;addition de l’accu avec le résultat de la 

;multiplication 
         RPT #14             ;décalage de 15 bits  
         SFR       
         
         AND #0fffch  
         LDP #0 
         SACL DXR             ;sortie vers le registre de transmission 
         RETE                 ; fin du programme 
 
 
* 
; 
; TRANSMIT INTERRUPT SERVICE ROUTINE 
; 
TRANSMIT: 
         RETE 

ANNEXE V : Variation du taux de modulation 
 
 
 

Pour m=0.01 

 
 
 
 

Pour m=0.25 

Pour m=0.8 

 
 

 
 

Pour m=0.99 

ANNEXE VI : Modification du signal informatif 
 

Signal modulé avec signal info à 500 Hz , m=0.25. 
 
 
 

 
 
 

Signal modulé avec signal info à 1000 Hz , m=0.25. 

ANNEXE VII : Variation de la fréquence porteuse 
 

Signal porteur à 3.34 KHz 

 
 
 

Signal porteur à 2.4 KHz 

Signal porteur à 1.8 KHz 
 
 

 
 

Signal porteur à 1.2 KHz 

